vJass



Далее следует свободный перевод ReedMe, в котором описаны все новые возможности, которые дает нам vJass.

Возможность объявлять глобальные переменные из любой части кода


Код в редакторе написанный на vJass:

Код:

// Стандартные генерируемые WE глобальные переменные:
globals
endglobals
//...
function MyFunc takes nothing returns nothing
endfunction
globals
    integer my_int=88
endglobals

Будет скомпилирован на Jass так:

Код:

// Стандартные генерируемые WE глобальные переменные:
globals
    integer my_int=88
endglobals
//...
function MyFunc takes nothing returns nothing
endfunction


Уже только из-за этого стоит использовать vJass - ведь во-первых можно давать переменным имена, которые будут удобными в использовании, напрямую указывать их тип, а также использовать свою функцию для их инициализации (т.е. сразу дать им необходимы значения)

Библиотеки и Области



Тут пойдет речь о "ограничение" кода, а именно его отрезков, что позволит использовать одинаковые имена функций и наоборот унифицировать некоторые функции. Также в свойствах библиотеки можно указать как она использует в свою очередь другие библиотеки, что позволит избежать мороки с расположением функций в коде. Это особенно удобно, когда используется много систем разных авторов.

//хотя лично мне не показались выше описанные примеры прямо скажем очень крутыми, может быть я просто не нашел им еще достойного применения. 

Библиотеки



Библиотека (library) - отрезок кода, который будет объявлен независимо от его положения в коде сразу после объявление глобальных переменных и выше CustomScriptCode.

Имеет следующий синтаксис:

Код:

library <name>

 /*тело библиотеки*/

endlibrary


Также библиотеки могут ссылаться на другие за счет следующей конструкции:

Код:

library A uses B

endlibrary


Это значит, что в библиотеки А содержится функция, вызывающая функцию в библиотеке В, следовательно библиотека В будет размещена выше по коду.

Пример на vJass:

Код:

library A requires B

 function SomeFunc takes nothing returns nothing
  call SomeFuncMain()

 endfunction
endlibrary
library B

 function SomeFuncMain takes nothing returns nothing
  call DoSomething()

 endfunction
endlibrary


На Jass Это выглядит как:

Код:

//library B:
 function SomeFuncMain takes nothing returns nothing
  call DoSomething()

 endfunction
//library B ends
//library A:
 function SomeFunc takes nothing returns nothing
  call SomeFuncMain()

 endfunction
//library A ends


Место uses может быть применено requires или needs, все три обозначения компилируются одинаково.
Также возможна следующая конструкция:

Код:

library A uses B, C, D


Это значит что библиотека А будет объявлена ниже B, C и D.

Важно отметить, что если библиотека А будет ссылаться на библиотеку В, а В на А это приведет к синтаксической ошибке.

Самоинициализация библиотек



Рассмотрим случай, когда мы написали какую либо систему, и поместили ее в библиотеку. Но для ее работы нам необходимо запустить ее инициализацию, к примеру присвоить значения глобальным переменным, создать некоторые объекты и т.д.

для этого используется следующая конструкция:

Код:

library A initializer InitA

endlibrary


Это значит, что в функцию main() (для тех кто не знает она запускается при загрузке карты) будет включен запуск функции InitA в отдельном потоке.

Как это выглядит на vJass:

Код:

ibrary A initializer InitA uses B

 function InitA takes nothing returns nothing
  set udg_i=i+1
 endfunction
endlibrary
library B initializer InitB

globals
 integer udg_i=0
endglobals
 function InitB takes nothing returns nothing
  set udg_i=2
 endfunction
endlibrary


Это будет скомпилированно в Jass'е как:

Код:

globals
 integer udg_i=0
endglobals
//library B:
 function InitB takes nothing returns nothing
  set udg_i = 2
 endfunction
//library B ends
//library A:
 function InitA takes nothing returns nothing
  set udg_i = 1
 endfunction
//library A ends
...

function main takes nothing returns nothing
...

 call ExecuteFunc("InitB")

 call ExecuteFunc("InitA")

 call InitGlobals()

endfunction


Замечу, что инициализация библиотек объявлена выше инициализации глобальных переменных.

Области



Области (scope) - по сути те же библиотеки, с различием в том, что они будут объявлены не вверху кода, а на своем реальном месте, не могут быть инициализированы, и не могут использовать другие библиотеки (хотя на самом деле это не все различия - прим. ADOLF)

Их использование аналогично использованию библиотек:

Код:

scope A

 /*тело области*/

endscope
Приватные конструкции



Возможен вариант, когда при импорте нескольких систем, написанных разными авторами имена некоторых функций и переменных могут быть одинаковыми.
Конечно можно перерыть весь код и потратив уйму сил и времени разобраться, что к чему относиться (хотя можно и авто заменой пройтись - прим. ADOLF)

Но есть вариант, когда можно объявить функцию или переменную как приватную (я бы назвал ее уникальной - прим. ADOLF)

Пример на vJass:

Код:

library A

globals
 private integer i=0
endglobals
private function MyFunc takes nothing returns nothing
 set i=0
endfunction
endlibrary
library B

globals
 private integer i=1
endglobals
private function MyFunc takes nothing returns nothing
 set i=1
endfunction
endlibrary


Фактически это может вызвать ошибку, но префикс private сделает данные обьекты уникальными, и ограничит их использование до использования только в данной библиотеки.

На Jass это будет выглядит так:

Код:

globals
 integer A9__i=0
 integer B9__i=1
endglobals
//library A:
function A9__MyFunc takes nothing returns nothing
 set A9__i = 0
endfunction
//library A ends
//library B:
function B9__MyFunc takes nothing returns nothing
 set B9__i = 1
endfunction
//library B ends


Т.е. к имени объекта будет добавлен префикс <library/scope nane>+<randonInt>+<__> 
Внутри данной конструкции обращать к объектам нужно без префикса.

Но предположим нам необходимо чтобы к некоторым объектам можно было обратиться "снаружи" библиотеки/области? Для этого используются публичные (универсальные - прим. ADOLF) префиксы, о которых и пойдет речь ниже.

Публичные конструкции



Для того, что бы сделать конструкцию публичной (универсальной - прим. ADOLF) - к ней надо прибавить префикс public.

Пример на vJass:

Код:

scope A

  public function MyFunc takes nothing returns nothing
  endfunction
  function InSide takes nothing returns nothing
    call MyFunc()

    call A_MyFunc()

  endfunction
 endscope
 function OutSide takes nothing returns nothing
   call A_MyFunc()

 endfunction


Что бы использовать публичный объект вне его библиотеки/области необходимо указать перед его именем префикс <library/scope name>+<_>.

Вышеуказанный код на скомпилированный на Jass будет выглядит так:

Код:

// scope A begins
 function A_MyFunc takes nothing returns nothing
 endfunction
 function InSide takes nothing returns nothing
  call A_MyFunc()

  call A_MyFunc()

 endfunction
// scope A ends
    function OutSide takes nothing returns nothing
         call A_MyFunc()

    endfunction


Обратите внимание, что обращаться к публичному объекту внутри конструкции можно с префиксом и без, снаружи только с.

Также стоит быть осторожным при использовании ExecuteFunc т.к. имена объектов будут изменены в зависимости от префикса private и public.

"Над" и "Под" конструкции



При написание какой либо весьма объемной системы может понадобиться использование конструкций типа scope.
В vJass можно помещать области в область и в библиотеку, но нельзя помещать библиотеку в какую либо конструкцию.

В принципе надеюсь вы уже поняли принцип работы данных конструкций, ниже я поясню принцип наследования имен.

Вот vJass код:

Код:

library A

 scope B

  public function MyFunc takes nothing returns nothing
  endfunction
 endscope
 public function AnotherMyFunc takes nothing returns nothing
 endfunction
endlibrary
function outside takes nothing returns nothing
 call A_B_MyFunc()

 call A_AnotherMyFunc()

endfunction


Поясню немного этот момент - при использование публичных (универсальных) объектов к ним стоит добавлять следующий префикс <library1/scope1 name>+<_>+<library2/scope2 name>+<_>+...

Также возможен вариант при размещения в разных областях подобластей с аналогичными именами:

Код:

scope A

 scope B

 endscope
endscope
scope C

 scope B

 endscope
endscope
Доступ к префиксам конструкции



Предположим, необходимо запустить приватную (уникальную) функцию в отдельной области в новом потоке. И, что бы узнать какой именно будет присвоен префикс к объектам в области можно воспользоваться конструкцией SCOPE_PREFIX/SCOPE_PRIVATE.

SCOPE_PREFIX - вернет в виде строки префикс конструкции для публичных (универсальных) объектов.

SCOPE_PRIVATE - вернет в виде строки префикс конструкции для приватных (унивкальных) объектов.

Пример на vJass:

Код:

scope A

 private function FuncA takes nothing returns nothing
 endfunction
 public function FuncB takes nothing returns nothing
 endfunction
 function FuncC takes nothing returns nothing
  call ExecuteFunc(SCOPE_PRIVATE+"FuncA")

  call ExecuteFunc(SCOPE_PREFIX+"FuncB")

 endfunction
endscope


Будет совершенно правильно скомпилирован на Jass:

Код:

// scope A begins
 function A5__FuncA takes nothing returns nothing
 endfunction
 function A_FuncB takes nothing returns nothing
 endfunction
 function FuncC takes nothing returns nothing
  call ExecuteFunc("A5__FuncA")

  call ExecuteFunc("A_FuncB")

 endfunction
// scope A ends
Ключевое слово


Вобщем стоит упомянуть и эту возможность, однако принцип ее работы будет описан ниже. Код на vJass:

Код:

scope A

 private keyword B

 private function A takes integer i returns integer
  if i!=0then
   return B.evaluate(i-1)

  endif
  return 0
 endfunction
 private function B takes integer i returns integer
  if i>99then
   return A(i-15)

  endif
  return 0
 endfunction
endscope


Будет скомпилирован на Jass как:

Код:

globals
//JASSHelper struct globals:
trigger array st___prototype2

integer f__result_integer

integer f__arg_integer1

endglobals
function sc___prototype2_execute takes integer i,integer a1 returns nothing
 set f__arg_integer1=a1

 call TriggerExecute(st___prototype2[ i])

endfunction
function sc___prototype2_evaluate takes integer i,integer a1 returns integer
 set f__arg_integer1=a1

 call TriggerEvaluate(st___prototype2[ i])

 return f__result_integer

endfunction
// scope A begins
 function A1__A takes integer i returns integer
  if i != 0 then
   return sc___prototype2_evaluate(1,i - 1)

  endif
  return 0
 endfunction
 function A1__B takes integer i returns integer
  if i > 99 then
   return A1__A(i - 15)

  endif
  return 0
 endfunction
// scope A ends
DeBug мод
Код:

function MyFunc takes nothig returns nothing
    call DoSomething1()

    debug call DoSomething 2()

endfunction


Если включен DeBug мод, то при проверки карты из редактора сработают обе функции, в противном случае только первая.

Текстовые вставки в код (TextMacro)



Данная структура позволит генерировать код, используя строчные переменные, которые могут внедряться в любую часть кода. Но, это всё статично, т.е. делаться только при сохранение карты в WE в игре таких чудес творить ясно что нельзя.

Для объявления textmacro используется следующая конструкция:

Код:

//! textmacro MyNameTextMacro takes string_var1, string_var2, string_var3,...

 /*тело вставки*/

//! endtextmacro


Важно отметить, что //! не комментарий, а именно обозначение конструкции.

Что бы вставить тот текст, который находиться в соответствующем textmacro в код необходимо написать:

Код:

//! runtextmacro MyNameTextMacro("<string1>", "<string1>", "<string1>",...)

Без вызова textmacro код, находящийся в нем будет проигнорирован. При его вызове текст, находящийся в textmacro будет вставлен в код в месте вызова с указанными аргументами.

Немного запутанно, на примере думаю будет понятней:

Код:

//! textmacro t takes type, name
function MyFunc$name$ takes $type$ returns nothing
endfunction
//! endtextmacro
//! runtextmacro t("integer i", "Int")
//! runtextmacro t("real r", "Real")


Будет скомпилировано на Jass как:

Код:

//textmacro instance: t("integer i", "Int")
function MyFuncInt takes integer i returns nothing
endfunction
//end of: t("integer i", "Int")
//textmacro instance: t("real r", "Real")
function MyFuncReal takes real r returns nothing
endfunction
//end of: t("real r", "Real")


Также textmacro может вообще не брать аргументы:

Код:

//! textmacro MyText
 call DoNothing()

//! endtextmacro
function MyFunc takes nothing returns nothing
 //! runtextmacro MyText()
endfunction


Данная конструкция совместима с библиотеками (library) и областями (scope).

Структуры (struct)



Структуры - очень удобная вещь, которая позволяет связать несколько переменных, передать их значения в другую функцию и т.д. Очень сильное средство в создание качественной тригерной магии.

Структура



Структура (struct) - связка некоторых переменных, которая используется тоже как переменная, с помощью которой можно получить доступ к включаемых в нее переменных. Обьявляется как:

Код:

struct <name>

 /*тело структуры*/

endstruct
local <struct name> <name>=<struct name>.create()

 set <name>.x=...

 //обращение к переменной x в структуре <name>
call <struct name>.destroy(<name>)


Сама по себе переменная-структура является также integer переменной.

Так это понять сложно, думаю наглядный примет тут уместнее:

Код:

struct cord

 real x

 real y

 real z

endstruct
function MyFunc takes nothing returns nothing
 local cord A=cord.create()

 set A.x=5
 set A.y=8
 set A.z=3
   call BJDebugMsg(R2S(A.x)+" "+R2S(A.y)+" "+R2S(A.z))

 call cord.destroy(A)

endfunction


На Jass это будет скомпилировано как:

Код:

globals
//JASSHelper struct globals:
integer si__cord_F=0
integer si__cord_I=0
integer array si__cord_V

real array s__cord_x

real array s__cord_y

real array s__cord_z

endglobals
//Generated allocator of cord
function s__cord__allocate takes nothing returns integer
 local integer this=si__cord_F

    if (this!=0) then
        set si__cord_F=si__cord_V[this]

    else
        set si__cord_I=si__cord_I+1
        set this=si__cord_I

    endif
    if (this>8190) then
        return 0
    endif
    set si__cord_V[this]=-1
 return this

endfunction
//Generated destructor of cord
function s__cord_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__cord_V[this]!=-1) then
        return
    endif
    set si__cord_V[this]=si__cord_F

    set si__cord_F=this

endfunction
...

function MyFunc takes nothing returns nothing
 local integer A=s__cord__allocate()

 set s__cord_x[A]=5
 set s__cord_y[A]=8
 set s__cord_z[A]=3
   call BJDebugMsg(R2S(s__cord_x[A]) + " " + R2S(s__cord_y[A]) + " " + R2S(s__cord_z[A]))

 call s__cord_destroy(A)

endfunction
"Над" и "Под" структуры



Структура может включать в себя другие структуры:

Код:

struct pair

 integer x=1
 integer y=2
endstruct
function pair_sum takes pair A, pair B returns pair

 local pair C=pair.create()

 set C.x=A.x+B.x

 set C.y=A.y+B.y

 return C

endfunction
struct pairpair

 pair x=0 

 pair y=0
 //вы не можете использовать pair.create() и  фактически должны использовать значения, присвоенные при инициализации.
endstruct
function testpairs takes nothing returns nothing
 local pairpair A=pairpair.create()

 local pair x

 //уже две переменные.
 set A.x=pair.create()

 set A.y=pair.create()

 //создаем две структуры, а фактически инициализируем 4 переменные -
 //A.x.x=1 ; A.x.y=2 ; A.y.x=1 ; A.y.y=2 
 set x=A.y 

 //мы сохраняем A.y в другую переменную, типа struct
 //что бы потом ее удалить.
 set A.y=pair_sum(A.x,A.y) 

 //и теперь спокойно помещаем сюда другую структуру.
 //вспомним, что pair_sum возращает новую структуру, в которой
 //содержиться сумма значение переменных аналогичных ей по строению двух других структур.
 call BJDebugMsg(I2S(A.y.x)+" : "+I2S(A.y.y)) //обратите внимание на пути обращения к структурам.
 //незабываем уничтожать ненужные структуры.
 call A.x.destroy()

 call A.y.destroy()

 call A.destroy()

 call x.destroy()

endfunction


Думаю тут все ясно, ведь структура - это переменная, поэтому она может быть помещена в другую и т.д.

Глобальные, приватные (уникальные) и публичные (универсальные) структуры



Структура может быть глобальной:

Код:

globals

 MyStuct str=0

endglobals


Имеется возможность использовать структуру как приватную (уникальную) и публичную (универсальную):

Код:

scope Sc

    public struct A

        integer i

    endstruct
    globals
        A x

        public A b

    endglobals
    public function test takes nothing returns nothing
        local A a=A.create()

        set b = A.create()

        set b.i = 3
        call a.destroy()

    endfunction
endscope
function test takes nothing returns nothing
 local Sc_A e=Sc_A.create()

    set e.i=6
    call e.destroy()

endfunction
Методы (method)



Методы (method) - внутриструктурные функции, имеющие простой доступ к переменным в структуре, предназначеные для их изменения. Имеют следующий синтаксис:

Код:

struct MyStr

integer my_int

method <name> takes nothing returns nothing
 /*тело метода*/

endmethod
endstruct


Достаточна простая в понимание конструкция, на vJass выглядит так:

Код:

struct Position

 real x

 real y

 real z

method move takes real nx, real ny, real nz returns nothing
 set this.x=this.x+nx

 set this.y=this.y+ny

 set this.z=this.z+nz

endmethod
endstruct
function MyFunc takes nothing returns nothing
 local Position A=Position.create()

 call A.move(1, 2, 3)

 call Position.destroy(A)

endfunction


На Jass:

Код:

globals
//JASSHelper struct globals:
integer si__Position_F=0
integer si__Position_I=0
integer array si__Position_V

real array s__Position_x

real array s__Position_y

real array s__Position_z

endglobals
//Generated allocator of Position
function s__Position__allocate takes nothing returns integer
 local integer this=si__Position_F

    if (this!=0) then
        set si__Position_F=si__Position_V[this]

    else
        set si__Position_I=si__Position_I+1
        set this=si__Position_I

    endif
    if (this>8190) then
        return 0
    endif
    set si__Position_V[this]=-1
 return this

endfunction
//Generated destructor of Position
function s__Position_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__Position_V[this]!=-1) then
        return
    endif
    set si__Position_V[this]=si__Position_F

    set si__Position_F=this

endfunction
...

function s__Position_move takes integer this, real nx, real ny, real nz returns nothing
 set s__Position_x[this]=s__Position_x[this] + nx

 set s__Position_y[this]=s__Position_y[this] + ny

 set s__Position_z[this]=s__Position_z[this] + nz

endfunction
function MyFunc takes nothing returns nothing
 local integer A=s__Position__allocate()

 call s__Position_move(A, 1 , 2 , 3)

 call s__Position_destroy(A)

endfunction


Как видно из кода, в метод добавляется еще один аргумент, который указывает в какой именно структуре стоит произвести изменения. Для обращения к переменной структуры используется префикс <this.> либо просто <.>

Обращаться к методу можно через call <Структура>.<имя метода> (<аргументы>).

Одна структура может иметь множество методов:

Код:

struct Some

 integer a

 integer b

method increase takes nothing returns nothing
 set this.a=this.a+1
 set this.b=this.b+1
endmethod
method decrease takes nothing returns nothing
 set this.a=this.a-1
 set this.b=this.b-1
endmethod
endstruct
function MyFunc takes nothing returns nothing
 local Some A=Some.create()

 call A.increase()

 call A.decrease()

 call A.destroy()

endfunction
Приватные и Публичные переменные в структурах



В структурах можно использовать как приватные(уникальные), так и публичные (универсальные) переменные, однако в случае использование первых их область применения - методы, входящие в состав структуры:

Код:

struct MyStr

 integer a

 public integer b

 private integer c

method MyMethod takes nothing returns nothing
 set this.a=0
 set this.b=0
 set this.c=0
endmethod
endstruct
function MyFunc takes nothing returns nothing
 local MyStr A=MyStr.create()

  set A.a=0
  set A.b=0
  //set A.c=0 - синтаксическая ошибка
 call A.destroy()

endfunction
Статические методы



Статический (static) метод не берет аргумент this следовательно в нем нельзя обращаться к переменной через this.

Но, статический метод может использовать приватные (уникальные) переменные структуры, в которой он обьявлен.

Статический метод может быть использован для альтернативного создания структуры или создания структуры с аргументами:

Код:

struct MyStr

 integer a=0
 public integer b=0
 private integer c=0
 method MyMethodA takes nothing returns nothing
  set this.a=0
 endmethod
 static method alt takes nothing returns MyStr

  local MyStr A=MyStr.create()

   set A.a=1
   set A.b=1
   set A.c=1
  return A

 endmethod
 method MyMethodB takes nothing returns nothing
  set this.a=0
 endmethod
endstruct
function OutSide takes nothing returns nothing
 local MyStr A=MyStr.alt()

 call A.destroy()

endfunction


Это будет скомпилированно как:

Код:

globals
//JASSHelper struct globals:
integer si__MyStr_F=0
integer si__MyStr_I=0
integer array si__MyStr_V

integer array s__MyStr_a

integer array s__MyStr_b

integer array s__MyStr_c

endglobals
//Generated allocator of MyStr
function s__MyStr__allocate takes nothing returns integer
 local integer this=si__MyStr_F

    if (this!=0) then
        set si__MyStr_F=si__MyStr_V[this]

    else
        set si__MyStr_I=si__MyStr_I+1
        set this=si__MyStr_I

    endif
    if (this>8190) then
        return 0
    endif
    set s__MyStr_a[this]=0
    set s__MyStr_b[this]=0
    set s__MyStr_c[this]=0
    set si__MyStr_V[this]=-1
 return this

endfunction
//Generated destructor of MyStr
function s__MyStr_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__MyStr_V[this]!=-1) then
        return
    endif
    set si__MyStr_V[this]=si__MyStr_F

    set si__MyStr_F=this

endfunction
...

 function s__MyStr_MyMethodA takes integer this returns nothing
  set s__MyStr_a[this]=0
 endfunction
 function s__MyStr_alt takes nothing returns integer
  local integer A=s__MyStr__allocate()

   set s__MyStr_a[A]=1
   set s__MyStr_b[A]=1
   set s__MyStr_c[A]=1
  return A

 endfunction
 function s__MyStr_MyMethodB takes integer this returns nothing
  set s__MyStr_a[this]=0
 endfunction
function OutSide takes nothing returns nothing
 local integer A=s__MyStr_alt()

 call s__MyStr_destroy(A)

endfunction


Статический метод может быть вызван снаружи структуры через <struct name>.<method name>(arguments)

Вот более наглядный пример разных созданий структур:

Код:

struct Some

 integer a

 static method altcreate takes integer a returns Some

  local Some A=Some.allocate() //allocate() - даст структуре уникальный ИД
   set A.a=a

  return A

 endmethod
endstruct
function test takes nothing returns nothing
 local Some A=Some.altcreate(0)

 local Some B=Some.create()

 call A.destroy()

 call B.destroy()

endfunction


На Jass выглядит как:

Код:

globals
//JASSHelper struct globals:
integer si__Some_F=0
integer si__Some_I=0
integer array si__Some_V

integer array s__Some_a

endglobals
//Generated allocator of Some
function s__Some__allocate takes nothing returns integer
 local integer this=si__Some_F

    if (this!=0) then
        set si__Some_F=si__Some_V[this]

    else
        set si__Some_I=si__Some_I+1
        set this=si__Some_I

    endif
    if (this>8190) then
        return 0
    endif
    set si__Some_V[this]=-1
 return this

endfunction
//Generated destructor of Some
function s__Some_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__Some_V[this]!=-1) then
        return
    endif
    set si__Some_V[this]=si__Some_F

    set si__Some_F=this

endfunction
...

 function s__Some_altcreate takes integer a returns integer
  local integer A=s__Some__allocate() //allocate() - даст структуре уникальный ИД
   set s__Some_a[A]=a

  return A

 endfunction
function test takes nothing returns nothing
 local integer A=s__Some_altcreate(0)

 local integer B=s__Some__allocate()

 call s__Some_destroy(A)

 call s__Some_destroy(B)

endfunction


Статичные методы берут столько же аргументов, сколько указанно в них (в отличии от обычных методов, которые дополнительно берут this - прим. ADOLF) - поэтому они могут быть использованы как обычные функции. Что бы получить доступ к имени функции - статичного метода необходимо добавить префикс <struct name.>

Вот пример:

Код:

struct MyStr

 method A takes nothing returns nothing
 endmethod
 static method B takes nothing returns nothing
 endmethod
endstruct
function OutSide takes nothing returns nothing
 local trigger a=CreateTrigger()

 local trigger b=CreateTrigger()

 //call TriggerAddAction(a, function MyStr.A) - приведет к синтаксической ошибке
 //т.к. реально метод А берет аргумент this
 call TriggerAddAction(b, function MyStr.B)

endfunction
Метод - уничтожитель (onDestroy method)



Метод, обьявленый внутри структуры и имеющий имя onDestroy будет автоматически вызван при уничтожение данной структуры. На vJass это будет выглядить так:

Код:

struct MyStr

 method onDestroy takes nothing returns nothing
 endmethod
endstruct
function OutSide takes nothing returns nothing
 local MyStr A=MyStr.create()

 call A.destroy()

endfunction


Устроено вообщето данная фича не очень то хорошо: при уничтожение структуры запускается тригер через функцию TriggerEvalute (т.е. данная функция запустит "условие тригера" а при инициализации тело метода onDestroy и будет помещенно в условие тригера, сопоставленного структуре), и через глобальную переменную передается значение this, причем действие тригера произойдет в его условие. Вот как вышеизложеный код будет скомпилирован:

Код:

//JASSHelper struct globals:
integer si__MyStr_F=0
integer si__MyStr_I=0
integer array si__MyStr_V

trigger st__MyStr_onDestroy

integer f__arg_this

endglobals
//Generated method caller for MyStr.onDestroy
function sc__MyStr_onDestroy takes integer this returns nothing
    set f__arg_this=this

    call TriggerEvaluate(st__MyStr_onDestroy)

endfunction
//Generated allocator of MyStr
function s__MyStr__allocate takes nothing returns integer
 local integer this=si__MyStr_F

    if (this!=0) then
        set si__MyStr_F=si__MyStr_V[this]

    else
        set si__MyStr_I=si__MyStr_I+1
        set this=si__MyStr_I

    endif
    if (this>8190) then
        return 0
    endif
    set si__MyStr_V[this]=-1
 return this

endfunction
//Generated destructor of MyStr
function sc__MyStr_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__MyStr_V[this]!=-1) then
        return
    endif
    set f__arg_this=this

    call TriggerEvaluate(st__MyStr_onDestroy)

    set si__MyStr_V[this]=si__MyStr_F

    set si__MyStr_F=this

endfunction
...

 function s__MyStr_onDestroy takes integer this returns nothing
 endfunction
//Generated destructor of MyStr
function s__MyStr_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__MyStr_V[this]!=-1) then
        return
    endif
    call s__MyStr_onDestroy(this)

    set si__MyStr_V[this]=si__MyStr_F

    set si__MyStr_F=this

endfunction
function OutSide takes nothing returns nothing
 local integer A=s__MyStr__allocate()

 call s__MyStr_destroy(A)

endfunction
...

//Struct method generated initializers/callers:
function sa__MyStr_onDestroy takes nothing returns nothing
local integer this=f__arg_this

endfunction
function jasshelper__initstructs4410241 takes nothing returns nothing
    set st__MyStr_onDestroy=CreateTrigger()

    call TriggerAddCondition(st__MyStr_onDestroy,Condition( function sa__MyStr_onDestroy))

endfunction


Создатели предпологали использовать данный метод для уничтожения во первых под структур (если их неуничтожать - то они рано или поздно переполняться и станут неработоспособными) а также для удаления обьектов, которые использует структура и которые больше ненужны. Выведение в отдельный метод по идеи должно облегчить данную операцию, т.к. все находиться в одной структуре.

Но, мое личное мнение, что нормальный програмист тщательно проверяет код, так что такие меры безопастности излишни, а также использование тригера тоже не очень хорошо скажется быстродействии алгоритма, что говорит против использования onDestroy method.

Интерфейсы и структуры



Если рассматривать структуру дальше как новый тип обьектов, можно предположить, что некоторые стандартные обьекты могут иметь аналогичные свойсва (т.е. координаты, жизнь и т.д.)

С помощью интерфейсов (interfaces) мы можем использовать общие переменные и методы для разных типов структур - по сути при обьеденение нескольких структур интерфейсом они станут лишь дополнением к основной структуре - интерфейсу.

Синтаксис следующий:

Код:

interface <name>

 /*тело интерфейса*/

endinterface
struct MyStruct extends MyInterface

endstruct


Вот пример:

Код:

interface A

 integer a

endinterface
struct B extends A

 integer b

endstruct
struct C extends A

 integer c

endstruct
function Test takes nothing returns nothing
 local B b=B.create()

 local C c=C.create()

  set b.a=1
  set b.b=2
  set c.a=3
  set c.c=4
  call BJDebugMsg(I2S(B(b).a)+" "+I2S(b.b)+" "+I2S(C(c).a)+" "+I2S(c.c))

endfunction


Скомпилированно это будет как:

Код:

globals
//JASSHelper struct globals:
constant integer si__A=1
integer si__A_F=0
integer si__A_I=0
integer array si__A_V

integer array si__A_type

trigger array st__A_onDestroy

integer array s__A_a

constant integer si__B=2
integer array s__B_b

constant integer si__C=3
integer array s__C_c

integer f__arg_this

endglobals
//Generated destructor of A
function sc__A_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__A_V[this]!=-1) then
        return
    endif
    set f__arg_this=this

    call TriggerEvaluate(st__A_onDestroy[si__A_type[this]])

    set si__A_V[this]=si__A_F

    set si__A_F=this

endfunction
//Generated allocator of C
function s__C__allocate takes nothing returns integer
 local integer kthis

 local integer this=si__A_F

    if (this!=0) then
        set si__A_F=si__A_V[this]

    else
        set si__A_I=si__A_I+1
        set this=si__A_I

    endif
    if (this>8190) then
        return 0
    endif
    set si__A_type[this]=3
    set kthis=this

    set si__A_V[this]=-1
 return this

endfunction
//Generated allocator of B
function s__B__allocate takes nothing returns integer
 local integer kthis

 local integer this=si__A_F

    if (this!=0) then
        set si__A_F=si__A_V[this]

    else
        set si__A_I=si__A_I+1
        set this=si__A_I

    endif
    if (this>8190) then
        return 0
    endif
    set si__A_type[this]=2
    set kthis=this

    set si__A_V[this]=-1
 return this

endfunction
...

function Test takes nothing returns nothing
 local integer b=s__B__allocate()

 local integer c=s__C__allocate()

  set s__A_a[ b]=1
  set s__B_b[ b]=2
  set s__A_a[c]=3
  set s__C_c[c]=4
  call BJDebugMsg(I2S(s__A_a[(b)]) + " " + I2S(s__B_b[ b]) + " " + I2S(s__A_a[(c)]) + " " + I2S(s__C_c[c]))

endfunction


Помните, что при обьеденение нескольких структур они будут использовать общий код их уничтожения, а также общие переменные для определения свободного слота, что делает максимальный общий размер структур обьедененных одним интерфейсом 8190, в то время при использование обчных структур их максимальный размер будет 8190 у каждой.

Код:

interface A

endinterface
struct B extends A

endstruct
struct C extends A

endstruct
function Func takes A a, B b, C c returns nothing
endfunction


В вышеизложенном коде показан пример передачи различных типов структур в качестве аргумента в другую функцию, однако реально же в коде будет передаваться integer. Вот как это будет скомпилированно:

Код:

function Func takes integer a, integer b, integer c returns nothing
endfunction


Поэтому всетаки стоит следить за тем, с каким именно типом структур вы оперируете. Хотя их можно и проверять:

Код:

interface A

endinterface
struct B extends A

endstruct
struct C extends A

endstruct
function Func takes A a, B b, C c returns nothing
 if a.getType()==B.typeid then
  //DoSomething
 elseif a.getType()==C.typeid then
  //DoNothing
 endif
endfunction


Реализованно это за счет того, что для каждой структуры/интерфейса создается одна integer константа, которая является ее индексом:

Код:

globals
...

constant integer si__A=1
constant integer si__B=2
constant integer si__C=3
...

endglobals
...

function Func takes integer a,integer b,integer c returns nothing
 if si__A_type[a] == si__B then
  //DoSomething
 elseif si__A_type[a] == si__C then
  //DoNothing
 endif
endfunction


Также тут нам пригодится способ обращаться к переменной в структуре следующим образом:

Код:

struct A

 integer i

endstruct
function Func takes nothing returns nothing
 local A a=A.create()

 set a.i=0
 set A(a).i=0
endfunction


Структуры, обьедененный интерфейсом также могут иметь общие методы:

Код:

interface printable

  method toString takes nothing returns string
 endinterface
 struct singleint extends printable

  integer v

  method toString takes nothing returns string
   return I2S(this.v)

  endmethod
 endstruct
 struct intpair extends printable

  integer a

  integer b

  method toString takes nothing returns string
   return "("+I2S(this.a)+","+I2S(this.b)+")"
  endmethod
 endstruct
 function printmany takes printable a, printable b, printable c returns nothing
  call BJDebugMsg( a.toString()+" - "+b.toString()+" - "+c.toString())

 endfunction
 function test takes nothing returns nothing
  local singleint x=singleint.create()

  local singleint y=singleint.create()

  local intpair z=intpair.create()

  set x.v=56
  set y.v=12
  set z.a=45
  set z.b=12
  call printmany(x,y,z)

 endfunction
Операторы



Операторы (Operator overloading) - методы, для использования которых используется обозначение [], []= и < >.

Первый тип оператора это [], который берет один integer аргумент и возращает что либо:

Код:

struct A

 integer i=0
 method operator [] takes integer i returns integer
  return .i+GetRandomInt(0, i)

 endmethod
endstruct
function test takes nothing returns nothing
 local A a=A.create()

 local integer i=0
  set i=a[5]

  call BJDebugMsg(I2S(i))

endfunction


Будет скомпилировано как:

Код:

function s__A__getindex takes integer this,integer i returns integer
 return s__A_i[this] + GetRandomInt(0 , i)

endfunction
function test takes nothing returns nothing
 local integer a=s__A__allocate()

 local integer i=0
  set i = s__A__getindex(a,5)

  call BJDebugMsg(I2S(i))

endfunction


Далее идет оператор []=, который берет integer аргумент и еще какой либо аргумент, но невозращает ничего:

Код:

struct A

 integer i=0
 method operator []= takes integer i, string s returns nothing
  set .i=.i+StringLength(s)+i

 endmethod
endstruct
function test takes nothing returns nothing
 local A a=A.create()

 local integer i=0
  set a[2]="omg"
  call BJDebugMsg(I2S(a.i))

endfunction


Данный оператор уже работает за счет тригера:

Код:

globals
//JASSHelper struct globals:
constant integer si__A=1
integer si__A_F=0
integer si__A_I=0
integer array si__A_V

integer array s__A_i

trigger st__A__setindex

integer f__arg_integer1

string f__arg_string1

integer f__arg_this

endglobals
//Generated method caller for A._setindex
function sc__A__setindex takes integer this,integer i,string s returns nothing
    set f__arg_this=this

    set f__arg_integer1=i

    set f__arg_string1=s

    call TriggerEvaluate(st__A__setindex)

endfunction
//Generated allocator of A
function s__A__allocate takes nothing returns integer
 local integer this=si__A_F

    if (this!=0) then
        set si__A_F=si__A_V[this]

    else
        set si__A_I=si__A_I+1
        set this=si__A_I

    endif
    if (this>8190) then
        return 0
    endif
    set s__A_i[this]=0
    set si__A_V[this]=-1
 return this

endfunction
//Generated destructor of A
function s__A_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__A_V[this]!=-1) then
        return
    endif
    set si__A_V[this]=si__A_F

    set si__A_F=this

endfunction
//library B:
function test takes nothing returns nothing
 local integer a=s__A__allocate()

 local integer i=0
  call sc__A__setindex(a,2, "omg")

  call BJDebugMsg(I2S(s__A_i[a]))

endfunction
//library B ends
...

function sa__A__setindex takes nothing returns boolean
local integer this=f__arg_this

local integer i=f__arg_integer1

local string s=f__arg_string1

  set s__A_i[this]=s__A_i[this] + StringLength(s) + i

   return true

endfunction
function jasshelper__initstructs2889244 takes nothing returns nothing
    set st__A__setindex=CreateTrigger()

    call TriggerAddCondition(st__A__setindex,Condition( function sa__A__setindex))

endfunction


И еще существует также оператор <, который берет какой либо аргумент и возращает boolean:

Код:

struct A

 integer i

 integer a

 method operator < takes A a returns boolean
  return this.i+this.a<a.i+a.a

 endmethod
endstruct
function Func takes nothing returns nothing
 local A a=A.create()

 local A b=A.create()

  set a.i=1
  set a.a=2
  set b.i=2
  set b.a=2
  if a<b then
   call BJDebugMsg("It work!")

  endif
endfunction


Данный оператор также устроен через тригер:

Код:

globals
//JASSHelper struct globals:
constant integer si__A=1
integer si__A_F=0
integer si__A_I=0
integer array si__A_V

integer array s__A_i

integer array s__A_a

trigger st__A__lessthan

integer f__arg_integer1

integer f__arg_this

boolean f__result_boolean

endglobals
//Generated method caller for A._lessthan
function sc__A__lessthan takes integer this,integer a returns boolean
    set f__arg_this=this

    set f__arg_integer1=a

    call TriggerEvaluate(st__A__lessthan)

 return f__result_boolean

endfunction
//Generated allocator of A
function s__A__allocate takes nothing returns integer
 local integer this=si__A_F

    if (this!=0) then
        set si__A_F=si__A_V[this]

    else
        set si__A_I=si__A_I+1
        set this=si__A_I

    endif
    if (this>8190) then
        return 0
    endif
    set si__A_V[this]=-1
 return this

endfunction
//Generated destructor of A
function s__A_destroy takes integer this returns nothing
    if this==null then
        return
    elseif (si__A_V[this]!=-1) then
        return
    endif
    set si__A_V[this]=si__A_F

    set si__A_F=this

endfunction
function Func takes nothing returns nothing
 local integer a=s__A__allocate()

 local integer b=s__A__allocate()

  set s__A_i[a]=1
  set s__A_a[a]=2
  set s__A_i[ b]=2
  set s__A_a[ b]=2
  if sc__A__lessthan(a,b) then
   call BJDebugMsg("It work!")

  endif
endfunction
...

//Struct method generated initializers/callers:
function sa__A__lessthan takes nothing returns boolean
local integer this=f__arg_this

local integer a=f__arg_integer1

set f__result_boolean= s__A_i[this] + s__A_a[this] < s__A_i[a] + s__A_a[a]

   return true

endfunction
function jasshelper__initstructs3228091 takes nothing returns nothing
    set st__A__lessthan=CreateTrigger()

    call TriggerAddCondition(st__A__lessthan,Condition( function sa__A__lessthan))

endfunction


Также операторы, как и простые методы могут быть обьявлены в интерфейсах.

Вообще какого либо преимущества перед обычними методами я тут невижу, так что лично я против их использования, т.к. большинство из них сделанно через тригер, что дает совершенно ненужную нагрузку (тоже можно реализовать и через обычный метод, но гораздо оптимальней).

Принцип работы и применение



Теперь я немного обьясню суть ее работы. Структура - это массив переменных (хотя обозначается на vJass как не массив - прим. ADOLF), следственно в структурах нельзя обьявлять массивы.

Структура при ее инициализации возращает нам свободный элемент массива, и обозначает его как использумый, и по уничтожению освобождает его, и обозначает как свободный (хотя реально алгоритм создания/удаления структуры куда интересней, думаю вам самим будет интересно разобраться как он устроен).

Однако, структуры могут быть рассмотренны как введение новых типов обьектов, характеристики которых хранятся в переменных. Впринципе выделение свободной ячейки массива может быть рассмотрено как выделение нового хендла к созданному вновь обьекту.

Сейчас поясню:

Код:

struct MyUnit

 real x

 real y

 real hp

 real mp

 real ms

 integer lvl

 ...

endstruct


Таким образомы мы можем создавать "пустой" обьект (как скажем trigger или timer), или при помощи статических методов создавать его сразу с характеристиками, заданными при создании (unit, rect)

Более того, мы можем имитировать native функции (т.е. функции для работы с каким либо типом обьетов) которые смогут менять его характеристики. Это все конечно реализуется с помощью методов:

Код:

struct MyRect

 real xmax

 real xmin

 real ymax

 real ymin

 method IncreaseMyRect takes nothing returns nothing
  set .xmax=.xmax+25
  set .ymax=.ymax+25
  set .xmin=.xmin+25
  set .ymin=.ymin+25
 endmethod
endstruct


Конечно, это будет только видомостью обьетка, т.к. движок вара с ним работать не будет, но реально же возможностей native функций достаточно, что бы влияние обьекта стало равно, или приблизилось к стандартным типам. Также далее будет расказанно, как использовать массивы в структурах, что даст еще больше возможностей.

Дальше я приведу еще один код на vJass:

Код:

struct pair

    integer x=1
    integer y=2
endstruct
function pair_sum takes pair A, pair B returns pair

 local pair C=pair.create()

    set C.x=A.x+B.x

    set C.y=A.y+B.y

 return C

endfunction
function testpairs takes nothing returns nothing
 local pair A=pair.create()

 local pair B=pair_sum(A, A)

 local pair C=pair_sum(A,B)

    call BJDebugMsg(I2S(C.x)+" : "+I2S(C.y))

    //Незабывайте удалять структуры, которые больше не нужны
    call B.destroy()

    call C.destroy()

    call pair.destroy(A)

endfunction
Выведет на экран "3 : 6"


Вот, значение переменных в структуре может быть задано изначально, в таком случае данная переменная при создание структуры примет указанное значение. 

Структуру по ее ненадобности стоит уничтожать, в противном случае после заполнения 8190 элементов она начнет возращать 0.
Уничтожать ее можно двумя методами:

Код:

struct A

 integer i

endstruct
function MyFunc takes nothing returns nothing
 local A a=A.create()

 local A b=A.create()

 call a.destroy()

 call A.destroy(b)

endfunction


Работает это одинаково.

Теперь ближе к повседневному применению: предположим мы пишем какую либо тригерную магию, в которой через кешь (или аналогичную систему на массивах, к примеру "ХАТ") вешаются какие либо значения.

Использование структур полностью меняет подход: мы вешаем на наш такймер только структуру (ее integer Значение) и через нее получаем доступ ко всем обьектам сохраненным в переменной структуры.
